

**tsfilter bw** — Butterworth time-series filter

|                             |                                      |                                |                                      |
|-----------------------------|--------------------------------------|--------------------------------|--------------------------------------|
| <a href="#">Description</a> | <a href="#">Quick start</a>          | <a href="#">Menu</a>           | <a href="#">Syntax</a>               |
| <a href="#">Options</a>     | <a href="#">Remarks and examples</a> | <a href="#">Stored results</a> | <a href="#">Methods and formulas</a> |
| <a href="#">References</a>  | <a href="#">Also see</a>             |                                |                                      |

## Description

`tsfilter bw` uses the Butterworth high-pass filter to separate a time series into trend and cyclical components. The trend component may contain a deterministic or a stochastic trend. The stationary cyclical component is driven by stochastic cycles at the specified periods.

See [\[TS\] `tsfilter`](#) for an introduction to the methods implemented in `tsfilter bw`.

## Quick start

Use the Butterworth filter for `y` and obtain cyclical component `ct` using `tsset` data

```
tsfilter bw ct=y
```

As above, but filter periods larger than 20

```
tsfilter bw ct=y, maxperiod(20)
```

As above, and save the trend component in the variable `trendvar`

```
tsfilter bw ct=y, maxperiod(20) trend(trendvar)
```

As above, and save gain and angular frequency with the names `gain` and `angle`

```
tsfilter bw ct=y, maxperiod(20) trend(trendvar) gain(gain angle)
```

Set the order of the filter to 4 and save the trend component in the variable `trendvar`

```
tsfilter bw ct=y, order(4) trend(trendvar)
```

Use Butterworth filter for variables `y1`, `y2`, and `y3` to obtain cyclical components with prefix `cycl`

```
tsfilter bw cycl* = y1 y2 y3
```

Note: The above commands can also be used to apply the filter separately to each panel of a panel dataset when a *panelvar* has been specified by using [tsset](#) or [xtset](#).

## Menu

Statistics > Time series > Filters for cyclical components > Butterworth

## Syntax

Filter one variable

```
tsfilter bw [type] newvar = varname [if] [in] [, options]
```

Filter multiple variables, unique names

```
tsfilter bw [type] newvarlist = varlist [if] [in] [, options]
```

Filter multiple variables, common name *stub*

```
tsfilter bw [type] stub* = varlist [if] [in] [, options]
```

*options*

Description

| <i>options</i>  | Description   |
|---|---|
| Main  |   |
| <u>maxperiod</u> (#)  | filter out stochastic cycles at periods larger than # |
| <u>order</u> (#)  | set the order of the filter; default is order(2)      |
| Trend   |   |
| <u>trend</u> ( <i>newvar</i>   <i>newvarlist</i>   <i>stub*</i> ) | save the trend component(s) in new variable(s)        |
| Gain  |   |
| <u>gain</u> ( <i>gainvar</i> <i>anglevar</i> )                    | save the gain and angular frequency                   |

You must `tsset` or `xtset` your data before using `tsfilter`; see [\[TS\] tsset](#) and [\[XT\] xtset](#).  
*varname* and *varlist* may contain time-series operators; see [\[U\] 11.4.4 Time-series varlists](#).

## Options

### Main

`maxperiod`(#) filters out stochastic cycles at periods larger than #, where # must be greater than 2. By default, if the units of the time variable are set to daily, weekly, monthly, quarterly, half-yearly, or yearly, then # is set to the number of periods equivalent to 8 years; otherwise, the default value is `maxperiod(32)`.

`order`(#) sets the order of the Butterworth filter, which must be an integer. The default is `order(2)`.

### Trend

`trend`(*newvar* | *newvarlist* | *stub\**) saves the trend component(s) in the new variable(s) specified by *newvar*, *newvarlist*, or *stub\**.

### Gain

`gain`(*gainvar* *anglevar*) saves the gain in *gainvar* and its associated angular frequency in *anglevar*. Gains are calculated at the *N* angular frequencies that uniformly partition the interval  $(0, \pi]$ , where *N* is the sample size.

## Remarks and examples

We assume that you have already read [TS] [tsfilter](#), which provides an introduction to filtering and the methods implemented in `tsfilter bw`, more examples using `tsfilter bw`, and a comparison of the four filters implemented by `tsfilter`. In particular, an understanding of gain functions as presented in [TS] [tsfilter](#) is required to understand these remarks.

`tsfilter bw` uses the Butterworth high-pass filter to separate a time-series  $y_t$  into trend and cyclical components:

$$y_t = \tau_t + c_t$$

where  $\tau_t$  is the trend component and  $c_t$  is the cyclical component.  $\tau_t$  may be nonstationary; it may contain a deterministic or a stochastic trend, as discussed below.

The primary objective is to estimate  $c_t$ , a stationary cyclical component that is driven by stochastic cycles within a specified range of periods. The trend component  $\tau_t$  is calculated by the difference  $\tau_t = y_t - c_t$ .

Although the Butterworth high-pass filter implemented in `tsfilter bw` has been widely applied by macroeconomists and engineers, it is a general time-series method and may be of interest to other researchers.

Engineers have used Butterworth filters for a long time because they are “maximally flat”. The gain functions of these filters are as close as possible to being a flat line at 0 for the unwanted periods and a flat line at 1 for the desired periods; see [Butterworth \(1930\)](#) and [Bianchi and Sorrentino \(2007, 17–20\)](#). (See [TS] [tsfilter](#) for an introduction to gain functions.)

The high-pass Butterworth filter is a two-parameter filter. The `maxperiod()` option specifies the maximum period; the stochastic cycles of all higher periodicities are filtered out. The `maxperiod()` option sets the location of the cutoff period in the gain function. The `order()` option specifies the order of the filter, which determines the slope of the gain function at the cutoff frequency.

For a given cutoff period, the slope of the gain function at the cutoff period increases with filter order. For a given filter order, the slope of the gain function at the cutoff period increases with the cutoff period.

We cannot obtain a vertical slope at the cutoff frequency, which is the ideal, because the computation becomes unstable; see [Pollock \(2000\)](#). The filter order for which the computation becomes unstable depends on the cutoff period.

Among economists, the high-pass Butterworth filter is commonly used for investigating business cycles. [Burns and Mitchell \(1946\)](#) defined business cycles as stochastic cycles in business data corresponding to periods between 1.5 and 8 years. For this reason, the default value for `maxperiod()` is the number of periods in 8 years, if the time variable is formatted as daily, weekly, monthly, quarterly, half-yearly, or yearly; see [D] [format](#). The default value for `maxperiod()` is 32 for all other time formats.


For each variable, the high-pass Butterworth filter estimate of  $c_t$  is put in the corresponding new variable, and when the `trend()` option is specified, the estimate of  $\tau_t$  is put in the corresponding new variable.

`tsfilter bw` automatically detects panel data from the information provided when the dataset was `tsset` or `xtset`. All calculations are done separately on each panel. Missing values at the beginning and end of the sample are excluded from the sample. The sample may not contain gaps.

### Example 1: Estimating a business-cycle component

In this and the subsequent examples, we use `tsfilter bw` to estimate the business-cycle component of the natural log of the real gross domestic product (GDP) of the United States. Our sample of quarterly data goes from 1952q1 to 2010q4. Below we read in and plot the data.


```
. use http://www.stata-press.com/data/r15/gdp2
(Federal Reserve Economic Data, St. Louis Fed)
. tsline gdp_ln
```



The series is nonstationary. [Pollock \(2000\)](#) shows that the high-pass Butterworth filter can estimate the components driven by the stochastic cycles at the specified frequencies when the original series is nonstationary.

Below we use `tsfilter bw` to filter `gdp_ln` and use `pergram` (see [\[TS\] pergram](#)) to compute and to plot the periodogram of the estimated cyclical component.

```
. tsfilter bw gdp_bw = gdp_ln
. pergram gdp_bw, xline(.03125 .16667)
```



`tsfilter bw` used the default value of `maxperiod(32)` because our sample is of quarterly data. In the periodogram, we added vertical lines at the natural frequencies corresponding to the conventional [Burns and Mitchell \(1946\)](#) values for business-cycle components. `pergram` displays the results in natural frequencies, which are the standard frequencies divided by  $2\pi$ . We use option `xline()` to draw vertical lines at the lower natural-frequency cutoff ( $1/32 = 0.03125$ ) and the upper natural-frequency cutoff ( $1/6 \approx 0.16667$ ).

If the filter completely removed the stochastic cycles at the unwanted frequencies, the periodogram would be a flat line at the minimum value of  $-6$  outside the range identified by the vertical lines.

The periodogram reveals two issues. First, it indicates that the default value of `order(2)` did not do a good job of filtering out the high-periodicity stochastic cycles, because there are too many points above  $-6.00$  to the left of the left-hand vertical line. Second, it reveals the high-pass nature of the filter, because none of the low-period (high-frequency) stochastic cycles have been filtered out.

We cope with these two issues in the remaining examples.


◀

## ► Example 2: Changing the order of the filter

In this example, we change the order of the filter so that it will remove more of the unwanted low-frequency stochastic cycles. As previously mentioned, increasing the order of the filter increases the slope of the gain function at the cutoff period.

For orders 2 and 8, we compute the filtered series, compute the gain functions, and label the gain variables. We also generate `ideal`, the gain function of the ideal band-pass filter at the frequencies `f`. Then we plot the gain function of the ideal band-pass filter and the gain functions of the high-pass Butterworth filters of orders 2 and 8.

```
. tsfilter bw gdp_bw2 = gdp_ln, gain(g1 a1)
. label variable g1 "BW order 2"
. tsfilter bw gdp_bw8 = gdp_ln, gain(g8 a8) order(8)
. label variable g8 "BW order 8"
. generate f = _pi*(_n-1)/_N
. generate ideal = cond(f<_pi/16, 0, cond(f<_pi/3, 1,0))
. label variable ideal "Ideal filter"
. twoway line ideal f || line g1 a1 || line g8 a8
```



As discussed in [TS] **tsfilter**, the gain function of the ideal filter is a square wave with a value of 0 at the frequencies corresponding to unwanted frequencies and a value of 1 at the desired frequencies. The vertical lines in the gain function of the ideal filter occur at  $\pi/16$ , corresponding to 32 periods, and at  $\pi/3$ , corresponding to 6 periods. (Given that  $p = 2\pi/\omega$ , where  $p$  is the period corresponding to frequency  $\omega$ , the frequency is given by  $2\pi/p$ .)

The distance between the gain function of the filter with order 2 and the gain function of the ideal band-pass filter at  $\pi/16$  is the root of the first issue mentioned at the end of [example 1](#). The filter with order 8 is much closer to the gain function of the ideal band-pass filter at  $\pi/16$  than is the filter with order 2. That both gain functions are 1 to the right of the vertical line at  $\pi/3$  reveals the high-pass nature of the filter.


◀

### ▶ Example 3: Removing the high-frequency component

In this example, we use a common trick to resolve the second issue mentioned at the end of [example 1](#). Keeping the trend produced by a high-pass filter turns that high-pass filter into a low-pass filter. Because we want to remove the high-frequency stochastic cycles still in the previously filtered series `gdp_bw8`, we need to run `gdp_bw8` through a low-pass filter. So we keep the trend produced by refiltering the previously filtered series.

To determine an order for the filter, we run the filter with `order(8)`, then with `order(15)`, and then we plot the gain functions along with the gain function of the ideal filter.


```
. tsfilter bw gdp_bwn8 = gdp_bw8, gain(gc8 ac8) order(8)
> maxperiod(6) trend(gdp_bwc8)
. label variable gc8 "BW order 8"
. tsfilter bw gdp_bwn15 = gdp_bw8, gain(gc15 ac15) order(15)
> maxperiod(6) trend(gdp_bwc15)
. label variable gc15 "BW order 15"
. twoway line ideal f || line gc8 ac8 || line gc15 ac15
```



We specified much higher orders for the filter in this example because the cutoff period is 6 instead of 32. (As previously mentioned, holding the order of the filter constant, the slope of the gain function at the cutoff period decreases when the period decreases.) The above graph indicates that the filter with `order(15)` is reasonably close to the gain function of the ideal filter.


Now we compute and plot the periodogram of the estimated business-cycle component.

```
. pergram gdp_bwc15, xline(.03125 .16667)
```



The graph indicates that the above applications of the Butterworth filter did a reasonable job of filtering out the high-periodicity stochastic cycles but that the low-periodicity stochastic cycles have not been completely removed.

Below we plot the estimated business-cycle component with recessions identified by the shaded areas.



## Stored results

`tsfilter bw` stores the following in `r()`:

### Scalars

`r(order)`            order of the filter  
`r(maxperiod)`        maximum period of stochastic cycles

### Macros

`r(varlist)`            original time-series variables  
`r(filterlist)`        variables containing estimates of the cyclical components  
`r(trendlist)`        variables containing estimates of the trend components, if `trend()` was specified  
`r(method)`            Butterworth  
`r(unit)`                units of time variable set using `tsset` or `xtset`

## Methods and formulas

`tsfilter bw` uses the computational methods described in [Pollock \(2000\)](#) to implement the filter.

[Pollock \(2000\)](#) shows that the gain of the Butterworth high-pass filter is given by

$$\psi(\omega) = \left[ 1 + \left\{ \frac{\tan(\omega_c/2)}{\tan(\omega/2)} \right\}^{2m} \right]^{-1}$$

where  $m$  is the order of the filter,  $\omega_c = 2\pi/p_h$  is the cutoff frequency, and  $p_h$  is the maximum period.

Here is an outline of the computational procedure that [Pollock \(2000\)](#) derived.

[Pollock \(2000\)](#) showed that the Butterworth filter corresponds to a particular model. Actually, his model is more general than the Butterworth filter, but `tsfilter bw` restricts the computations to the case in which the model corresponds to the Butterworth filter.

The model represents the series to be filtered,  $y_t$ , in terms of zero mean, covariance stationary, and independent and identically distributed shocks  $\nu_t$  and  $\varepsilon_t$ :

$$y_t = \frac{(1+L)^m}{(1-L)^m} \nu_t + \varepsilon_t$$

From this model, [Pollock \(2000\)](#) shows that the optimal estimate for the cyclical component is given by

$$\mathbf{c} = \lambda \mathbf{Q}(\mathbf{\Omega}_L + \lambda \mathbf{\Omega}_H)^{-1} \mathbf{Q}' \mathbf{y}$$

where  $\text{Var}\{\mathbf{Q}'(\mathbf{y} - \mathbf{c})\} = \sigma_\nu^2 \mathbf{\Omega}_L$  and  $\text{Var}\{\mathbf{Q}'\mathbf{c}\} = \sigma_\varepsilon^2 \mathbf{\Omega}_H$ . Here  $\mathbf{\Omega}_L$  and  $\mathbf{\Omega}_H$  are symmetric Toeplitz matrices with  $2m+1$  nonzero diagonal bands and generating functions  $(1+z)^m(1+z^{-1})^m$  and  $(1-z)^m(1-z^{-1})^m$ , respectively.

The parameter  $\lambda$  in this expression is a function of  $p_h$  (the maximum period of stochastic cycles filtered out) and the order of the filter:

$$\lambda = \{\tan(\pi/p_h)\}^{-2m}$$



The matrix  $\mathbf{Q}'$  in this expression is a function of the coefficients in the polynomial  $(1 - L)^d = 1 + \delta_1 L + \dots + \delta_d L^d$ :

$$\mathbf{Q}' = \begin{pmatrix} \delta_d & \dots & \delta_1 & 1 & \dots & 0 & 0 & \dots & 0 & 0 \\ \vdots & \ddots & \vdots & \vdots & \ddots & \vdots & \vdots & & \vdots & \vdots \\ 0 & \dots & \delta_d & \delta_{d-1} & \dots & 1 & 0 & \dots & 0 & 0 \\ 0 & \dots & 0 & \delta_d & \dots & \delta_1 & 1 & \dots & 0 & 0 \\ \vdots & & \vdots & \vdots & \ddots & & & \ddots & \vdots & \vdots \\ 0 & \dots & 0 & 0 & \dots & \delta_d & \delta_{d-1} & \dots & 1 & 0 \\ 0 & \dots & 0 & 0 & \dots & 0 & \delta_d & \dots & \delta_1 & 1 \end{pmatrix}^{(T-d) \times T}$$

It can be shown that  $\Omega_H = \mathbf{Q}'\mathbf{Q}$  and  $\Omega_L = |\Omega_H|$ , which simplifies the calculation of the cyclical component to

$$\mathbf{c} = \lambda \mathbf{Q} \{ |\mathbf{Q}'\mathbf{Q}| + \lambda (\mathbf{Q}'\mathbf{Q}) \}^{-1} \mathbf{Q}'\mathbf{y}$$

## References

- Bianchi, G., and R. Sorrentino. 2007. *Electronic Filter Simulation and Design*. New York: McGraw-Hill.
- Burns, A. F., and W. C. Mitchell. 1946. *Measuring Business Cycles*. New York: National Bureau of Economic Research.
- Butterworth, S. 1930. On the theory of filter amplifiers. *Experimental Wireless and the Wireless Engineer* 7: 536–541.
- Pollock, D. S. G. 1999. *A Handbook of Time-Series Analysis, Signal Processing and Dynamics*. London: Academic Press.
- . 2000. Trend estimation and de-trending via rational square-wave filters. *Journal of Econometrics* 99: 317–334.
- . 2006. Econometric methods of signal extraction. *Computational Statistics & Data Analysis* 50: 2268–2292.

## Also see

- [TS] **tsset** — Declare data to be time-series data
- [XT] **xtset** — Declare data to be panel data
- [TS] **tsfilter** — Filter a time-series, keeping only selected periodicities
- [D] **format** — Set variables' output format
- [TS] **tssmooth** — Smooth and forecast univariate time-series data